
  الخامسالفصل  5

  

  
  الواجھات، التفويض، والمواصفات

  
 تبقى لـدينا  ،Visual Basic .NETنحن على مشارف الانتهاء من تعلم أساسيات لغة البرمجة و

 بأساسيات لغة للإلماممجموعة من المواضيع المتفرقة والتي يتحتم على ذكرها من منطلق الشمولية 
  .Visual Basic .NETالبرمجة 

 من هذا الكتاب بالحديث عـن مواضـيع متعـددة كالواجهـات             الأولسأختم معك الجزء    
Interfaces       وطريقة تطبيق مبدأ تعدد الواجهات Polymorphism  ،      كما سأتطرق أيـضا إلـى

 والتي تعطيك مرونة كبيرة في استدعاء الإجـراءات، واخـتم           Delegatesالإجراءات المفوضة   
 لننهي بذلك مرحلة تعلـم أساسـيات لغـة    Attributesالفصل بالتلميح إلى موضوع المواصفات  

  .Visual Basic .NETالبرمجة 

  
  الواجھات

 هـي  Interfaceالواجهة . OOPة واجهة في لغات  لك ما المقصود بكلم أوضحفي البداية دعني    
  :مجموعة من الطرق والخصائص والأحداث التي تصف فئة معينة، فالفئة التالية

 
Class Person 
    Event Die() 
 
    Public Property Name() As String 
        … 
        … 
    End Property 
 
    Sub Move() 
        … 
        … 
    End Sub 
End Class 

. ()Move، وطريقة باسـم     Name، خاصية باسم    Dieتحتوي على واجهة متمثلة في حدث باسم        
سـلوكها أو   لـيس    أي( الأعضاء فقط وليس وظائفهـا       أسماءلاحظ ان كلمة الواجهة تطلق على       

  ).الشيفرات التي تحتويها
، وهـو   Polymorphismتعدد الواجهات    هو مبدأ    OOP في لغات    الأساسيةمن المبادئ   

 إلى ان تتشارك مجموعة فئات في     الأمر من واجهة تصفها، ليصل      أكثراحتواء الفئة الواحدة على     
 Same names but different مختلفـة  إنجـازات  متشابهة لكن أسماء: واجهات موحدة فنقول

implementation.  
، Polymorphismواجهات  العديد من الفوائد التي تجنيها من تطبيق واستخدام مبدأ تعدد ال          

، TextFile اختصار جمل الشرط في برنامجك، فتخيل ان لدينا فئة تمثل مستند نصي           أبرزهالعل  
، HTMLFile تمثل مستند صفحة ويـب       أخرى، وفئة   RTFFile تمثل مستند منسق     أخرىوفئة  

 ستجد  ،PictureFile تمثل صورة أخيرة، وفئة XMLFileوفئة رابعة تمثل مستند تنسيق البيانات  
 كل فئة من هذه الفئات سيختلف  تماما عن الفئة الأخرى، ولكن الواجهات التي               وإنجازان برمجة   

 أو حجم الملـف  FileNameتحتويها متشابهة ومتشاركة، فيمكن تعريف خاصية تمثل اسم الملف     
FileSize     وطريقة تؤدي إلى عملية الحفظ ،Save() أو الحذف Delete()يـع  ، ويمكن تطبيق جم

  .هذه الواجهات على الفئات المختلفة
فمثلاً لو قمت بتطوير إجراء يقوم بحفظ الملف النصي فستجعل الوسيطة تستقبل كائن مـن               

  :TextFileالنوع 
 
Sub SaveTextFile(ByVal FileObject As TextFile) 
    … 
    FileObject.Save 
    … 
End Sub 

  
  :من الملفاتونفس الشئ ستفعله مع الانواع الاخرى 

 
Sub SaveRTFFile(ByVal FileObject As RTFFile) 
    … 
    FileObject.Save 
    … 
End Sub 
 
Sub SaveHTMLFile(ByVal FileObject As HTMLFile) 
    … 
    FileObject.Save 
    … 
End Sub 

 الجمل الشرطية في ليس هذا فقط، بل أنه عند استدعاء إجراء الحفظ  ستجد أنه يتوجب عليك كتابة        
كل مرة تود فيها فعل ذلك للتحقق من نوع الملف، وعلى ضوء المقارنة تقوم باستدعاء فئة الملف                 

 عن جميع الكائنات التي تمثل الفئات المختلفـة         الإعلانالمعنية لتقوم بالحفظ، كما أن عليك أيضا        
  : إلى الإجراء المختصلإرسالها

 
Dim TextFileObject As New TextFile 
Dim RTFFileObject As New RTFFile 
Dim HTMLFileObject As New HTMLFile 
 
… 
… 
 
Select Case FileType 
    Case 1 
        TextFileObject.FileName = "xxxx" 
        SaveTextFile ( TextFileObject ) 
        … 
    Case 2 
        RTFFileObject.FileName = "xxxx" 
        RTFRTFFile ( RTFFileObject ) 
        … 
    Case 3 
        HTMLFileObject.FileName = "xxxx" 
        HTMLTextFile ( HTMLFileObject ) 
        … 
    … 
    … 
End Sub 

  
 مثلا، BinaryFile فئة جديدة لنوع معين من التسيقات       لإضافة لنفترض انه ظهرت الحاجة      والآن

 والـشروط التـي     والإعلانات والإجراءاتات  ءفإن ذلك يفرض عليك ان تضيف جميع الاستدعا       
  . البرنامج، وتقوم بتكرار كل ما فعلته في الفئات السابقةأنحاءستتعامل مع هذه الفئة في كل 

 هنا هو القيـام بتعريـف     هع تعدد الواجهات فإن الأمر مختلف تماماً، فكل ما سنفعل         ولكن م 
واجهة ال الكائنات شريطة احتوائه على      أنواع ونعرف إجراء يستقبل كل      IFileواجهة نسميها مثلا    

IFile:  
 
Sub SaveFile(ByVal FileObject As IFile) 
    … 
    FileObject.Save () 
    … 
End Sub 

، TextFile( التـي ذكرناهـا سـابقا        الأنـواع  يقتصر الإجراء السابق على اسـتقبال        لا
RTFFile  ،HMTLFile  ، ...والتي قـد تحتـوي   الأخرى الأنواعبل يمكنه ان يستقبل كل   ) الخ 
  .IFile على الواجهة الأياميوما من 

  
  بناء واجھة

، تحتوي على جميع IFile الاسم  لك الفكرة من الواجهات، لنبدأ ببناء واجهة تحملأوضحتبعد ان 
، Visual Basic .NETلتعريف واجهة فـي  .  استخدامهاالأخرىالأعضاء التي نود من الفئات 

  :Interface … End Interfaceاستخدم التركيب 
 
Interface IFile 
 خصائص '    
    Property FileName() As String 
    Property FileSize() As Integer 
    … 
 
 طرق '    
    Sub Save() 
    Sub Save(ByVal TargetFile As String) 
    Sub Delete() 
    … 
End Interface 

  
 شيفرات برمجية فهي مجـرد واصـفة   أيةكما ترى في التركيب السابق، الواجهة لا تحتوي على      

، End Sub ،End Function ووسيطات الأعضاء فقط، فلا تفكر في وضع عبارات كـ لأسماء
  .End Propertyأو 
  

    ملاحظة
  
 قبـل اسـم   I باستخدام الحـرف  OOPرى العرف سابقا عند مبرمجي      ج

ــو نفــس   ــة، وھ ــستندات  الأســلوبالواجھ ــه م ــذي تقترح ــع NET. ال  م
  . لكتابة اسم الواجھةPascalCaseاستخدام 

  
  

 اسـتخدام    ويمكنـك  Friend للواجهات، فهي افتراضيا     Visibilityبالنسبة لقابلية الرؤية    
  : كما تفعل مع الفئات تماماالأخرىمحددات الوصول 

' Friend 
Interface IFile 
    … 
End Interface 
 
' Public 
Public Interface IView 
    … 
End Interface 
 
Module Module1 
    ' Private 
    Private Interface ITool 
        … 
    End Interface 
 
    Sub Main() 
        … 
    End Sub 
End Module 

  
مع ذلك، لا يمكنك استخدام محددات الوصول مع أعضاء الواجهة، فجميع أعضاء الواجهات قابلية         

  :وأبدا دائما Publicرؤيتها 
 
Interface ITest 
 لا يمكن استخدام محددات الوصول '    
  الواجهةأعضاءمع  '    
    Public Sub MyMethod () 
    Private Sub YourMethod () 
    … 
    … 
End Interface 

  
  :Nestedيمكن للواجهات ان تكون متداخلة المزيد أيضا، 

 
Interface IView 
    Interface IWindow 
        … 
        … 
    End Interface 
 
    Interface IWEB 
        … 
        … 
    End Interface 
    … 
    … 
End Interface 

 الحقـول  أمـا هي التي يمكن كتابتها في الواجهات،    والأحداث ، الخصائص، اخيرا، الطرق 
  :فغير ممكنة

 
Interface IMyInterface 
    X As Inetger ' بودنا ولكن للاسف 
    … 
End Interface 

  
  تضمین الواجھة

  :Implementجوزة  فئة باستخدام الكلمة المحأي في هايمكنك تضمينبعد تعريفك للواجهة، 
 
Class TextFile 
    Implements IFile ' تضمين الواجهة 
    … 
    … 
End Class 

  
لا إ لتضمين الواجهة في الفئة الحالية هو تعريف جميع أعضاء الواجهـة و            الأساسيالشرط  

  :ستظهر لك رسالة خطأ
 
Class TextFile 
    Implements IFile 
 
    ' FileName ةتضمين الخاصی  
    Property FileName() As String Implements IFile.FileName 
        Get 
            … 
        End Get 
        Set(ByVal Value As String) 
            … 
        End Set 
    End Property 
 
    ' Delete() تضمين الطریقة 
    Sub Delete() Implements IFile.Delete 
        … 
    End Sub 
     
  الاخرىعضاءعلیك اضافة جمیع الأ '    
    … 
    … 
End Class 

وبدلا من تعريف أعضاء الواجهات يدويا، يمكنك الاستعانة بمحرر الشيفرات لعمل ذلك نيابة 
 عنك، حدد الواجهة من القائمة العلوية اليسرى ومن ثم اختر العضو المراد تعريفه مـن القائمـة                

  .)1-5شكل (العلوية اليمنى 
  

  
  . الاستعانة بمحرر الشيفرة لكتابة أعضاء الواجهة:1-5شكل 

  
عند تعريف أعضاء الواجهة، فلا بد من توافق وسيطات الطرق، الأحداث، والخصائص مع           
نفس الوسيطات الموجودة في الواجهة، ولابد من توافق نـوع القـيم التـي تعـود بهـا الـدوال           

Functions .اقـصد اسـتخدام الكلمـات    (زيد أيضا، قابلية القراءة والكتابـة   المReadOnly و 
WriteOnly (          تـضمينات  فال. للخصائص، لابد وان تكون مطابقة لتلك الموجودة فـي الواجهـة

  %:100 خاطئة  بالصفحة التاليةTestClassالموجودة في الفئة 
 
Interface IMyInterface 
 خاصیة للقرائة فقط '    
    ReadOnly Property BirthDate() As Date 
     
 دالة تعود بقیمة حرفیة '    
    Function MyMethod() As String 
End Interface 

Class TestClass 
    Implements IMyInterface 
 
 قابلیة القرائة والكتابة غير متوافقة '    
    Public WriteOnly Property BirthDate() As Date Implements _ 
        IMyInterface.BirthDate 
     
            Set(ByVal Value As Date) 
                … 
                … 
            End Set 
    End Property 
 
 القیمة التي تعود بها الدالة غير متوافقة '    
    Public Function MyMethod() As Integer Implements _  
        IMyInterface.MyMethod 
 
           … 
           … 
    End Function 
End Class 

  
بالنسبة للواجهات المتداخلة، فيمكنك تضمينها كما تفعل مع الواجهات العادية، ولكن لا تنسى 

  :أولاكتابة اسم الواجهة الحاضنة لها 
 
Interface IView 
    Interface IWindow 
        … 
    End Interface 
 
    Interface IWEB 
        … 
    End Interface 
    … 
End Interface 
 
Class TestClass 
    Implements IView.IWindow ' لا تنسى كتابة اسم الواجهة الحاضنة اولا 
 
    … 
    … 
End Class 

  
  : استخدمت الواجهة الحاضنةإناما 

 
Class TestClass2 
    Implements IView 
    … 
End Class 

 فقط، ولا  تعتقد للحظة ان الواجهات IViewفالمؤكد  ان الفئة السابقة لن تحتوي إلا على الواجهة    
  .الأخرىقد تم تضمينها هي ) IWeb و IWindow(ونة بها ضالمح

  : من واجهةأكثر، يمكن للفئة الواحدة ان تحتوي على أخيرا
 
Class ManyInterfaces 
    Implements IFile 
    Implements IView 
    Implements IView.IWindow 
    Implements IView.IWEB 
 
    … 
    … 
End Class 

  
عند تعارض اسماء الأعضاء في اكثر من واجهتين، فلابد من تغيير اسمها لحظة كتابتها في  

 و  IViewودة في الواجهتين الـسابقتين       موج ()MyMethodالفئة، فمثلا لو اكتشفت أن الطريقة       
IFileفيمكنك تمييزهما بوضع رقم ،:  

 
Class ManyInterfaces 
    Implements IFile 
    Implements IView 
    … 
    … 
 
    Sub MyMethod() Implements IFile.MyMethod 
        … 
    End Sub 
 
 لابد من تغیير اسم الطریقة '    
    Sub MyMethod2() Implements IView.MyMethod 
        … 
    End Sub 
    … 
    … 
End Class 

  
 لا يهتم باسم العضو بقدر ما يهتم Visual Basic .NETمن المثال السابق، يتضح لنا ان مترجم 

 لحظة تعريفه، لنستنتج من ذلك انه يمكننا تغيير         Implementsبالاسم الذي يلي الكلمة المحجوزة      
  :م الطرق السابقةاس

Class ManyInterfaces 
    Implements IFile 
    Implements IView 
    … 
    … 
 
    Sub MethodFromFile () Implements IFile.MyMethod 
        … 
    End Sub 
 
 
 
    Sub MethodFromView () Implements IView.MyMethod 
        … 
    End Sub 
    … 
    … 
End Class 

  
  الوصول إلى الواجھة

قد تم تضمين   ) الخ ...TextFile  ،RTFFile  ،HTMLFile(بافتراض ان فئات الملفات السابقة      
 اي قيمة له من النـوع  وإسناد IFile بها، فان باستطاعتنا تعريف متغير من النوع        IFileالواجهة  

TextFile ،RTFFile ،HTMLFile... الخ:  
 
Dim textFileObject As New TextFile() 
Dim file As IFile 
 
 ممكن عمل ذلك '
file = textFileObject 
file.FileName = "MyFile.TXT" 
… 
… 

  
  :IFileكما يمكن تعريف إجراء يستقبل وسيطة من النوع 

 
Sub DoSave(ByVal fileObject As IFile) 
    … 
    fileObject.Save () 
    … 
End Sub 

  
  :الخ ....TextFile ،RTFFile ،HTMLFile اي قيمة له من النوع رسالإلتتمكن من 

 
 
 
 

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


  الواجھات، التفويض، والمواصفات: الخامسالفصل   197  
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  
Dim htmlFileObject As New HMTLFile() 
… 
… 
DoSave (htmlFileObject) 
… 
… 

  
  وراثة الواجھات

 عند الأول من منظورين مختلفين، المنظور إليهاعند الحديث عن وراثة الواجهات، يمكننا ان ننظر 
  : تحتوي على واجهةةقاعدياشتقاق فئة 

 
Interface IMyInterface 
    Sub MyMethod() 
End Interface 
 
Class BaseClass 
    Implements IMyInterface 
 
    Sub MyMethod() Implements IMyInterface.MyMethod 
        … 
    End Sub 
End Class 

  
المضمونة بها   IMyInterface الواجهة   ، عليك معرفة ان    السابقة BaseClassعندما تشتق الفئة    
  :سيتم اشتقاقها أيضا

 
Class DerivedClass 
    Inherits BaseClass 
 
    … 
    … 
End Class 

  
  :BaseClass كما تتعامل مع الكائن DerivedClassلتتعامل مع الكائن من الفئة 

 
Dim DerivedObject As New DerivedClass() 
Dim X As IMyInterface 
 
X = DerivedObject 
 
 واجهة من الفئة القاعدیة '
X.MyMethod() 

اما المنظور الثاني فهو يتعلق بوراثة الواجهات، وهو توارث الواجهات فيما بينها باستخدام             
  : أيضاInheritsالكلمة المحجوزة 

 
Interface IBaseInterface 
    Sub MyMethodInBase() 
End Interface 
 
Interface IDerivedInterface 
    Inherits IBaseInterface 
 
    Sub MyMethodInDerived() 
End Interface 

  
 في اي فئة، فيتوجب عليك تعريف جميع الأعـضاء          IDerivedInterfaceعند تضمين الواجهة    

  : أيضاالقاعديةالتابعة للواجهة 
 
Class TestClass 
    Implements IDerivedInterface 
 
اجهة القاعدیةالو '      
    Sub MyMethodInBase() Implements IDerivedInterface.MyMethodInBase 
        … 
    End Sub 
 
 الواجهة المشتقة '    
    Sub MyMethodInDerived() Implements 
IDerivedInterface.MyMethodInDerived 
        … 
    End Sub 
End Class 

  
  NET Framework. عمل إطارواجھات من 
 وذكـرت حينهـا انـك       ،IDisposable تعرفنا على الواجهة     الفئات والكائنات في الفصل الثالث    

  : في الفئةDestructors لمحاكاة المهدمات ()Disposeتضمنها في الفئة لتعرف الإجراء 
 
Class TestClass 
    Implements IDisposable 
 
 
    Public Sub Dispose() Implements System.IDisposable.Dispose 
        … 
    End Sub 
    … 
    … 
End Class 

 NET Framework. عمل إطار والمعرفة في مكتبة فئات الأخرىتوجد مئات الواجهات 

 في هذا القسم عرض بضعة واجهات هامـة تـستخدمها فـي معظـم برامجـك        سأحاول أنيالا  
  .Visual Basic .NETومشاريعك بـ 

  
  IComparableة الواجھ

، <(عملية إجراء المقارنة بين عددين تعتمد على قيمة العدد، ويمكنك استخدام معاملات المقارنـة        
، حيث سيعتمد  Strings، ويمكن أيضا استخدام علامات المقارنة بين القيم الحرفية          )الخ= ...<،  >

 أو  ASCIIالمترجم في هذه الحالة على المقابـل الرقمـي للحـرف فـي جـداول المحـارف                  
UNICODEحسب الحالة :  

 
Dim X As Integer = 10 
Dim Y As String = "تركي" 
 
If X > 5 Then … 
If Y < "احمد" Then … 

  
 والتي تصممها بفئاتك الخاصة، فـلا يمكـن        الأخرى البيانات   أنواعولكن عند الحديث عن     

هي القيم   تحدد بنفسك ماتفأن Personتحديد ايهما اكبر وايهما اصغر، فلو عرفت كائنين من فئة           
 في الفئـة  IComparableالتي تريد الاعتماد عليها لحظة المقارنة، لذلك سنقوم بتضمين الواجهة  

كان الكائن الحالي اصغر من   اذا1- والتي تعود بالقيمة ()CompareToلتعرف طريقتها الوحيدة 
  :ن الحالي اكبر من الكائن المرسل ان كان الكائ1 في حالة تساوي الكائنين، أو 0الكائن المرسل، 

 
Class Person 
    Implements IComparable 
 
    Public Name As String 
 
    Function CompareTo(ByVal obj As Object) As Integer Implements _ 
        System.IComparable.CompareTo 
 
        Dim tempObj As Person = CType(obj, Person) 
     
 الكائن الحالي اصغر من الكائن المرسل '        
        If Me.Name < tempObj.Name Then 
            Return -1 
  من الكائن المرسلاكبرالكائن الحالي  '        
        ElseIf Me.Name > tempObj.Name Then 
            Return 1 

 الكائنان متساویان '        
        Else 
            Return 0 
        End If 
    End Function 
End Class 

  
لتحديد ناتج المقارنة بين الكائنين، قد  Nameفي المثال السابق، اعتمدت على خاصية الاسم   

  : مثلا في فئاتك الخاصةAge  رأيك وتعتمد على خاصية العمرأنتتغير 
 
 

Class Person 
    Implements IComparable 
    Public Age As Integer 
 
    Function CompareTo(ByVal obj As Object) As Integer Implements _ 
        System.IComparable.CompareTo 
 
        Dim tempObj As Person = CType(obj, Person) 
 
        ' Age الاعتماد على خاصیة العمر 
        If Me.Age < tempObj.Age Then 
            Return -1 
        ElseIf Me.Age > tempObj.Age Then 
            Return 1 
        Else 
            Return 0 
        End If 
    End Function 
End Class 

  
ة  الكائنات المنشئ  إرسال لذلك تستطيع    IComparable تحتوي على الواجهة     Person فئتنا   والآن

والتي تتطلـب الواجهـة    (الأخرى NET Framework.من هذه الفئة إلى جميع إجراءات فئات 
IComparable طبعا  .(   فمثلا الطريقةSort())    والتابعة للفئةArray (      يمكـن ان تـستقبل هـذا

  :النوع
 
 

Dim PersonObject(10) As Person 
Dim Counter As Integer 
 
For Counter = 0 To UBound(PersonObject) 
    PersonObject(Counter) = New Person() 
Next 
 
PersonObject(0).Age = 33 
PersonObject(1).Age = 50 
PersonObject(2).Age = 14 

… 
… 
 
 الطریقة Array.Sort تقبل اي وسیطة تحتوي '
' IComparable على الواجهة 
Array.Sort(PersonObject) 
 
For Counter = 0 To UBound(PersonObject) 
    ArabicConsole.WriteLine(PersonObject(Counter).Age) 
Next 

  
  ICloneableالواجھة 

 من مرة أكثر أخبرتك، فقد Reference Type Variablesعند التعامل مع المتغيرات المرجعية 
 تقوم وإنما، لآخراان عملية إسناد القيم بين متغيرين لا تقوم بإنشاء نسخة من المتغير وإسنادها إلى   

  :بجعل كلا المتغيرين يشيران إلى نفس الكائن
 
Dim Turki As New Person () 
Dim Ali As Person 
 
… 
… 
 
  نسخة جدیدة وانماإنشاءلا تقوم ب '
  نفس الكائنإلىجعل كلا المتغرین یشيران  '
Ali = Turki 

  
تعريف طريقتها الوحيدة  لICloneable تقوم بتضمين الواجهة NET.معظم الفئات في عالم 

Clone()     مثـال  هـذا  ، آخروالتي تقوم بنسخ الكائن وإنشاء نسخة جديدة منه وإسنادها إلى متغير
  :لتعريف الطريقة في فئة

 
Class Person 
    Implements ICloneable 
 
    Public Name As String 
    Public Age As Integer 
 
    Function Clone() As Object Implements ICloneable.Clone 
  كائن مؤقتإنشاء '        
        Dim tmpObject As New Person() 
 
  الكائنإلىنسخ قیم الكائن الحالي  '        
 المؤقت یدویا '        
        tmpObject.Name = Me.Name 
        tmpObject.Age = Me.Age 

  للتوالعودة بالكائن المنشئ '        
        Return tmpObject 
    End Function 
End Class 

  
حيث انه يمكن أن توجد متغيرات سـتاتيكية  (وبدلا من قيامك بإنشاء الكائن ونسخ خصائصه يدويا    

Static            فيمكنـك اسـتخدام الطريقـة       ) داخل الطرق والتي لن تـتمكن مـن نـسخها بنفـسك ،
MemberwiseClone()      وهي مشتقة من الفئة System.Object) للتـذكير بـان جميـع       أعود 

  ):System.Object مشتقة وراثيا من الفئة NET. في عالم يء وكل شوالأنواعالبيانات 
 
Class Person 
    Implements ICloneable 
 
    Public Name As String 
    Public Age As Integer 
 
    Function Clone() As Object Implements ICloneable.Clone 
 العودة بنسخة جدیدة من الكائن الحالي '        
        Return Me.MemberwiseClone() 
    End Function 
End Class 

  
  : يمكنك استخدامها بهذا الشكلوالآن

 
Dim Turki As New Person() 
Dim Ali As Person 
 
Turki.Name = "تركي" 
 
 نسخ الكائن '
Ali = CType(Turki.Clone, Person) 
 
Ali.Name = "علي" 
 
 تحقق ان الكائن الاول لم یتم تغیير قیمته '
ArabicConsole.WriteLine(Turki.Name) 

  
  

 بافتراض ان العبارة CTypeكما ترى في الشيفرة السابقة، اضطررت إلى استخدام المعامل 
Option Strict On ب ان الطريقـة   مفعلة في الملف الحالي أو على مستوى المشروع، والـسب

Clone()      تعود بقيمة من النوع Object .      قد يكـون اسـتخدام المعامـلCType   غيـر محبـذ 

 في الفئة واجعلها مخفية، ومن ثم ()Clone اسم الطريقة اقترح عليك تغييرلمستخدمي فئاتك، لذلك 
  : لتحل محلهاأخرىعرف طريقة 

 
Class Person 
    Implements ICloneable 
    … 
    … 
 
    Private Function PrivateClone() As Object Implements 
ICloneable.Clone 
        Return Me.Clone() 
    End Function 
 
    Public Function Clone() As Person 
        Return CType(Me.MemberwiseClone(), Person) 
    End Function 
End Class 

  
  :CType دون الحاجة لاستخدام المعامل ()Cloneفئة من استدعاء الطريقة هنا سيتمكن مستخدم ال

 
Ali = Turki.Clone 

  
، دعني أحذرك وأنبهك بان عمليات النسخ التي        ICloneableالواجهة  وقبل ان اختم فقرة     

توجد بها مشكلة خطيرة جدا جدا، ولم ) ()MemberwiseCloneباستخدام الطريقة (قمنا بها سابقا 
لذلك، لا تتبع الأسلوب السابق في نسخ       .  هنا لان حلها يتم عن طريق ما يعرف بالتسلسل         اعرضها

 Objectتسلـسل الكائنـات   الكائنات بمشاريعك الحالية حتى تـصل إلـى الفـصل التاسـع     
Serialization.اللهم إني قد بلغت، اللهم فاشهد .  

  
  IEnumerator و IEnumerableالواجھتان 

 علـى  For Each … Next انك تستطيع تطبيـق الحلقـة   لغة البرمجةي ذكرت في الفصل الثان
 بلغـة الواجهـات   الآن نتحـدث  أننـا ، وبما Collections أو المجموعات  Arraysالمصفوفات  
Interfaces ان الحلقة وأخبرك دقةً معك، أكثر أكون فدعني For Each يمكن تطبيقها أيضا على 

  .IEnumerator و IEnumerableم الواجهتين  البيانات الاخرى التي تدعأنواعجميع 
 جنبا إلى جنب لتمكين المبرمجين من       IEnumerator و   IEnumerableتعمل الواجهتان   

 مع فئاتك، وهذا مثال عملي يشرح كيفية استخدام هـاتين الـواجهتين،   For Eachاستخدام الحلقة 
جملة إلى كلمات، تحتوي هذه الفئـة        والتي تقوم بتقسيم ال    SplitStringوسنبدأ أولاً بتصميم الفئة     

  : والتي تمثل الجملة المراد فصل كلماتهاSentenceعلى الخاصية 

Class SplitString 
    Private currentPosition As Integer = 0 
 
الجملة المراد فصل كلماتهاخاصیة تمثل  '      
    Private m_Sentence As String 
 
    Property Sentence() As String 
        Get 
            Return m_Sentence 
        End Get 
        Set(ByVal Value As String) 
            m_Sentence = Value 
            Me.Reset() 
            End Set 
    End Property 
End Class 

  
  : في الفئة السابقةIEnumerator و IEnumerableالخطوة التالية هي تضمين الواجهتين 

 
 

Class SplitString 
    Implements IEnumerable 
    Implements IEnumerator 
    … 
    … 
End Class 

  
 ()GetEnumerator على طريقة واحـدة      إلا، فهي لا تحتوي     IEnumerableبالنسبة للواجهة   

يمة لاي كائن يحتوي على الواجهـة  ، وهي تعود بقFor Eachسيتم استدعائها بمجرد تنفيذ حلقة 
  : لتنفيذهاReturn Me الأمريكفي كتابة . IEnumerator الأخرى

 
 

Class SplitString 
    … 
    … 
    Private Function GetEnumerator() As IEnumerator Implements _  
        IEnumerable.GetEnumerator 
 
        Return Me 
    End Function 
End Class 

 الأولـى  فهي تحتوي على طريقتين وخاصية واحدة، الطريقة         IEnumeratorاما الواجهة   
، ويفترض For Each يتم استدعائها في كل دورة من دورات حلقة التكرار ()MoveNextتسمى 

  :شيء لم يتبقى اي إن False بقي عنصر تالي و إن Trueان تعود بالقيمة 
 

Class SplitString 
    … 
    … 
    Private Function MoveNext() As Boolean Implements 
IEnumerator.MoveNext 
        If Me.currentPosition > Me.Sentence.Length - 1 Then 
                Me.Reset() 
            Return False 
        Else 
            Return True 
        End If 
    End Function 
End Class 

  
 المؤشر الداخلي في الفئة إلى أول عنصر فيها    إعادة والغرض منها    ()Resetوالطريقة الثانية هي    

  :أخرى مرة For Eachحتى تتمكن من البدء باستخدام الحلقة 
 

Class SplitString 
    … 
    … 
    Private Sub Reset() Implements IEnumerator.Reset 
        Me.currentPosition = 0 
    End Sub 
End Class 

  
 والغرض منها العودة بالقيمة الحالية في الحلقة، لا تنسى       Currentاما الخاصية فهي تحمل الاسم      

  :ReadOnly فقط للقراءةان هذه الخاصية 
 

Class SplitString 
    … 
    … 
    Private ReadOnly Property Current() As Object Implements _  
        IEnumerator.Current 
 
        Get 
            Dim counter As Integer 
            Dim tmpLength As Integer = 0 

            For counter = Me.currentPosition To Me.Sentence.Length - 1 
                If Me.Sentence.Chars(counter) = " "c Then 
                    Exit For 
                Else 
                    tmpLength += 1 
                End If 
            Next 
            Current = Me.Sentence.Substring(Me.currentPosition, 
tmpLength) 
            Me.currentPosition += tmpLength + 1 
        End Get 
    End Property 
End Class 

  
 بمرونة كبيرة، لتتمكن من SplitString على الفئة السابقة For Each يمكنك تطبيق الحلقة والآن

  :كتابة شيئا مثل
 
 

Dim testObject As New SplitString() 
Dim x As String 
 
testObject.Sentence = "سیتم فصل كلمات هذه الجملة في سطور مستقلة" 
 
For Each x In testObject 
    ArabicConsole.WriteLine(x) 
Next 

  
  :مخرجات الشيفرة السابقة ستكون كالتالي

 
 سیتم
 فصل
 كلمات
 هذه
 الجملة
 في
 سطور
 مستقلة

    ملاحظة

  
 أوضـح  أن أردت ھـو تعلیمـي بحـت،       SplitString الفئة   إنشاءالھدف من   

، فـلا  IEnumerator و IEnumerableمن خلاله كیفیة تطبیق الواجھتین     
 NET.تحاول استخدامھا في برامجك الجدية، حیث يوفر لك اطار عمـل  

Framework وأسرع بكثیر لتقسیم النصوصأفضل فئات .  
، من IEnumerator و IEnumerableن ی حول تضمین الواجھتأخرىنقطة 

المفضل ألا تقوم بتـضمین ھـاتین الـواجھتین فـي فئـة واحـدة، فیفـضل              
ــضمین الو ــة تــ ــة    IEnumerableاجھــ ــنة، والواجھــ ــة حاضــ ــي فئــ  فــ

IEnumerator      ،وذلك لتـصمیم كـائني توجـه      في فئة محضونة بھاOOP 
  .أفضل

  

  
  التفويض
بتفويض عملية استدعاء الإجراء إلى متغير يشير إلى ذلـك          : Delegatesالتفويض  يقصد بكلمة   

جراءات، كما يختصر عليك الكثير  استدعاء الإأوقاتالإجراء، مما يعطيك مرونة كبيرة في اختيار 
  :من الشيفرات المصدرية والتي تستدعي هذه الإجراءات، فلو كان لدينا الإجراءات التالية

 
Sub MySub (ByVal X As Integer) 
    … 
    … 
End Sub 
 
Sub YourSub (ByVal X As Integer) 
    … 
    … 
End Sub 

  :إليهايمثل مؤشر  عن طريق متغير استدعائهافإنه بامكاننا 
 
' Dlgt الذي یشير له المتغيرجراءسیتم استدعاء الإ  
 كما سترسل الیه قیم الوسیطات '
Dlgt.Invoke (100) 

  
 وتمكن الإجراء مـن   اخرإلى إجراء كوسيطة   هذه المؤشرات    إرسالتستطيع  المزيد أيضا،   

  :استدعائها
 
Sub ShowError(ByVal delegatePointer As OneParameter) 
    delegatePointer.Invoke("حدث خطأ في العملیة") 
End Sub 

تفيدك الطريقة السابقة كثيرا في اختصار تكرار الشيفرات المتشابهة، حيـث ان الإجـراء              
، Console، فـي شاشـة   Message Boxالسابق قد يظهر رسالة الخطأ اما على شـكل نافـذة   

الخ، دون الحاجة ...، أو الكتابة في ملف )ان كنت تصمم موقعاً    (Browser عرضها على متصفح  
 -سترى تطبيقاً عملياً عليها قريباً-ميزة اخرى اكثر إثارة. لاعادة استدعاء الإجراء السابق من جديد

 Sharedمن داخل الإجراءات المـشتركة      ) أعضاء الفئات ( استدعاء إجراءات الفئات     إمكانيةهي  
  .دون مشاكل

 Functionمؤشـرات الـدوال   قد يخطـر ببالـك    ف++C\Cان كنت من مبرمجي لغة 
Pointers      بشكل مبدئي، مؤشرات الدوال .  والتي تمكنك من استدعاء الدوال عن طريق مؤشراتها
 ولكنها تختلف في Visual Basic .NET شبيهة بالإجراءات المفوضة في لغة ++C\C في لغة

 اكثـر  Visual Basic .NET ميكانيكية عملها وصيغ كتابتها، حيث ان الإجراءات المفوضة في
 وذلك لانها تحصر نوعية الإجراءات بحيث تتوافق مع      ++C\Cأماناً من مؤشرات الدوال في لغة       

 يمكن ان تتبع لإجراءات Visual Basic .NET، الإجراءات المفوضة في أخرىنقطة . وسيطاتها
مـا مؤشـرات   ، بينInstance Procedures أو تابعـة لفئـات   Static Proceduresستاتيكية 
  . لا يمكن تطبيقها الا مع الإجراءات الستاتيكية فقط++C\C في لغة تالإجراءا

  
  الإجراءات الستاتیكیة

 أو Subsسـواء كانـت   ( الإجـراءات  Static Proceduresاقصد بعبارة الإجراءات الستاتيكية 
Functions (     التي تعرفها في الفئاتClasses     بحيث تكون مشتركة ) مة المحجوزة  باستخدام الكل

Shared(        أو التي تعرفها في الوحدات البرمجية ،Modules .      ،عند تفويض الإجراء إلى مؤشـر
  :Delegatesعليك اولا القيام بتعريف نوع مؤشر الإجراء باستخدام الكلمة المحجوزة 

 
Module Module1 
  لابد ان یكونجراءتعریف نوع مؤشر الإ '    
مجیةعلى مستوى الوحدة البر '      
    Delegate Function OneParameter(ByVal X As Integer) As Integer 
     
    Sub Main() 
        … 
        … 
    End Sub 
    … 
    … 
End Module 

  :جرائين التاليين لدينا الإأنولنفترض الآن 
 

Function Abs(ByVal x As Integer) As Integer 
    If x < 0 Then 
        Return -x 
    Else 
        Return x 
    End If 
End Function 
 
Function Square(ByVal x As Integer) As Integer 
    Return x * x 
End Function 

  
الذي عرفناه سابقا باستخدام الكلمـة       (OneParameterونريد تفويضها إلى اي مؤشر من النوع        

  : بالطريقة التاليةفي اي وقت) Delegateالمحجوزة 
 

Sub Main () 
    ' Abs () جراءتفویض الإ  
    Dim Dlgt As New OneParameter(AddressOf Abs) 
    … 
    … 
End Sub 

  
 ()Invoke عن طريق هذا المؤشـر باسـتدعاء الطريقـة           ()Absالان يمكنك استدعاء الإجراء     
  :وارسال الوسيطات المطلوبة

 
Sub Main () 
    Dim Dlgt As New OneParameter(AddressOf Abs) 
 
جراءاستدعاء الإ '      
    ArabicConsole.WriteLine( Dlgt.Invoke(-5) ) ' 5 
End Sub 
 

باسـتخدام الكلمـة   ( في اي وقت بإسناد عنوانـه  ()Squareتستطيع الآن تفويض الإجراء الاخر     
لاستدعاء سـتتم بـنفس الطريقـة        السابق، وعملية ا   Dlgtإلى المؤشر   ) AddressOfالمحجوزة  
Invoke():  

 
Sub Main () 
    … 
    … 
    ' Square () جراءتفویض الإ  
    Dlgt = AddressOf Square 
    ArabicConsole.WriteLine( Dlgt.Invoke(-5) ) ' 25 
End Sub 

 لتـستدعي  OneParameterالمزيد أيضا، تستطيع تعريف إجراء يستقبل وسـيطة مـن النـوع       
  : الكائن المرسلإليها التي يشير تالإجراءا

 
Sub Main() 
    Dim Dlgt As New OneParameter(AddressOf Abs) 
    … 
    … 
    MySub(Dlgt) 
End Sub 
 
' OneParameter یستقبل وسیطة من النوعإجراء  
Sub MySub(ByVal delegatePointer As OneParameter) 
    ArabicConsole.WriteLine(delegatePointer.Invoke(-5)) 
End Sub 

  
 لك حقيقة استخدام الكلمـة المحجـوزة   لأوضحدعنا نرى الآن ماذا يحدث خلف الكواليس،      

Delegates كما في السطر الاول من المثال السابق عندما عرفت النوع OneParameter:  
 
Delegate Function OneParameter(ByVal X As Integer) As Integer 

  
 مشتقة وراثيا مـن     OneParameterتقنيا، قام مترجم اللغة بتحويل العبارة السابقة إلى فئة باسم           

، )System.Delegateوالتي هي أيضا مشتقة من الفئـة       (System.MulticastDelegateالفئة  
  :OneParameterاي يمكنك افتراض أن الشيفرة السابقة قد تم كتابتها لتعريف فئة 

 
فتراض، فلا تطبقه في الواقعهذا مجرد ا '  
Class OneParameter 
    Inherits System.MulticastDelegate 
    … 
    … 
End Class 

  
 ومن New إنشاء نسخة جديدة من الكائن اضطررنا لاستخدام الكلمة المحجوزة          أردنالذلك، عندما   

  : القيم لوسيطات المشيدإرسالثم 
 
Dim Dlgt As New OneParameter(AddressOf Abs) 

 إضـافية  مؤشر لكائن يحتوي على خـصائص وطـرق   Dlgtما أود الوصول إليه هو ان    
، كما يمكنك التعامل   ) والتي تستدعي الإجراء المشار اليه     ()Invoke وهي الطريقة    إحداهاذكرت  (

  :مع هذا الكائن كما تتعامل من كائنات الفئات التي تعرفها بنفسك لتتمكن من كتابة شيئاً مثل
 
Dim Dlgt As New OneParameter(AddressOf Abs) 
Dim X As OneParameter 
 
  قیم بين كائنينإسناد '
X = Dlgt 
 
ArabicConsole.WriteLine(X.Invoke(-5)) ' 5 

  
  إجراءات الفئات

 رغبت في تطبيق التفـويض      إنفي الفقرة السابقة طبقنا التفويض على الإجراءات الستاتيكية، اما          
 انك بحاجة إلى استخدام اسم الكـائن  إلا السابق الأسلوبءات الفئات فالعملية ستتم بنفس  على إجرا 

  :إلى المؤشر المفوض) AddressOfباستخدام ( عنوان الإجراء إرسالالذي يتبع له الإجراء عند 
 
Class TestClass 
    Sub TestMethod() 
        ArabicConsole.WriteLine("طریقة من الكائن") 
    End Sub 
End Class 
 
Module Module1 
    Delegate Sub NoParameter() 
 
    Sub Main() 
        Dim testObject As New TestClass() 
 
        ' testObject لاحظ هنا استخدام اسم الكائن 
        Dim dlgt As New NoParameter(AddressOf testObject.TestMethod) 
 
        dlgt.Invoke() 
    End Sub 
End Module 

  
باستخدام التفويض، يمكنك تخطي حاجز المنع من استدعاء إجـراءات الفئـة مـن داخـل         

 أنه لا يمكنك الفئات والكائناتالإجراءات المشتركة في الفئة، حيث وكما ذكرت في الفصل الثالث        
  :استدعاء طريقة في الفئة من داخل طريقة مشتركة في نفس الفئة

Class TestClass 
    Sub InstanceMethod() 
        … 
        … 
    End Sub 
 
    Shared Sub SharedMethod() 
 لا يمكنك عمل ذلك '        
        InstanceMethod() 
    End Sub 
End Class 

ام متغيرات عامـة     باستخد إما الأساليب بعشراتتستطيع تطبيق العملية السابقة باستخدام التفويض       
Global Variables المشتركةللإجراءات كوسيطات إرسالها أو :  

 
Class TestClass 
    Sub InstanceMethod() 
        ArabicConsole.WriteLine("تم استدعاء الطریقة") 
    End Sub 
 
    Shared Sub SharedMethod(ByVal methodPointer As NoParameter) 
        methodPointer.Invoke() 
    End Sub 
End Class 
 
Module Module1 
    Delegate Sub NoParameter() 
 
    Sub Main() 
        Dim testObject As New TestClass() 
        Dim dlgt As New NoParameter(AddressOf 
testObject.InstanceMethod) 
 
        ' InstanceMethod() دعاء الطریقةسیتم است  
        ' SharedMethod() من خلال الطریقة المشتركة 
        testObject.SharedMethod(dlgt) 
    End Sub 
End Module 

  
  محاكاة الأحداث

 التي تعرفها في    Events استخدمت المنطق البرمجي قليلا، فسرعان ما ستكتشف ان الأحداث           إن
  : التاليDie ولكن بصيغة مختلفة، فالحدث Delegates التفويض أسلوبالفئات تطبق 

 
Class Person 
    Event Die() 
 
    Sub Kill() 
        RaiseEvent Die() 
    End Sub 
End Class 

Module Module1 
    Sub Main() 
        Dim Turki As New Person() 
 
        AddHandler Turki.Die, AddressOf PeronHasDied 
 
        Turki.Kill() 
    End Sub 
 
    Sub PeronHasDied() 
        ArabicConsole.WriteLine("لقد توفي الشخص") 
    End Sub 
End Module 

  
، فأنت تقوم بتحديد عنوان الإجراء الـذي سـيتم تنفيـذه            Die تفويض متمثل في الاسم      إلاهو   ما

 الأحداث الـسابق وتعريـف      إطلاق صيغة   يمكنك أيضا الاستغناء عن   . AddHandlerباستخدام  
ومن ثم التصريح عن متغير في الفئة يحمل اسـم الحـدث            )  مثلا NoParameterباسم  (تفويض  

Die   هذا الحدث ستقوم باستدعاء الطريقة       إطلاق أردت، وان Invoke()      الأمـر   عوضـا عـن
RaiseEvent .        لإطـلاق ت التفويض   الشيفرة التالية هي نفس الشيفرة السابقة، ولكني هنا استخدم 

  :الحدث بدلاً من الصيغة المباشرة
 
Delegate Sub NoParameter() 
 
Class Person 
    Public Die As NoParameter 
 
    Sub Kill() 
        Die.invoke() 
    End Sub 
End Class 
 
Module Module1 
    Sub Main() 
        Dim Turki As New Person() 
 
        Turki.Die = New NoParameter(AddressOf PeronHasDied) 
 
        Turki.Kill() 
    End Sub 
 
    Sub PeronHasDied() 
        ArabicConsole.WriteLine("لقد توفي الشخص") 
    End Sub 
End Module 

  دمج التفويضات
 من تفويض في مؤشـر  أكثر دمج إمكانية الرائعة جدا في استخدام تقنية التفويض، هو        الأشياءمن  

.  من عصفور بحجر واحـد أكثر من إجراء باستدعاء واحد لتضرب     أكثرواحد بحيث يمكنك تنفيذ     
  :فمثلا، لنفترض ان لدينا الاجرائين التاليين

 
Sub MySub1() 
    ArabicConsole.WriteLine("الإجراء الاول") 
End Sub 
 
Sub MySub2() 
    ArabicConsole.WriteLine("الإجراء الثاني") 
End Sub 

  
-ونريد استدعاء الاجرائين باستخدام التفويض، فإنه يتوجب علينا استدعاء كل إجراء على حـدة               

  :()Invokeباستخدام الطريقة 
 
Delegate Sub NoParameter() 
 
sub Main () 
    Dim dlgt As New NoParameter(AddressOf MySub1) 
    Dim dlgt2 As New NoParameter(AddressOf MySub2) 
 
    dlgt.Invoke() 
    dlgt2.Invoke() 
End Sub 

  
 لتدمج اكثر مـن  ()Combineاما عند الدمج فلست بحاجة لكل هذا، حيث يكفي استخدام الطريقة          

 باستدعاء واحد   ()MySub2 و   ()MySub1تفويض في مؤشر واحد، لتتمكن من تنفيذ الاجرائين         
  :تستطيع فعل ذلك بإنشاء متغير ثالث للقيام بهذه العملية. ()Invokeفقط للطريقة 

 
Delegate Sub NoParameter() 
 
Sub Main() 
    Dim dlgt As New NoParameter(AddressOf MySub1) 
    Dim dlgt2 As New NoParameter(AddressOf MySub2) 
    Dim dlgt3 As NoParameter 
 
    dlgt3 = dlgt.Combine(dlgt, dlgt2) 
 
    ' MySub2 و MySub1 سیتم تنفیذ كلا الاجرائين 
    dlgt3.Invoke() 
End Sub 

  : الوسيطاتإرسال استخدام متغير واحد فقط لتنفيذ هذه العملية، بحيث يتم إنشاء الكائنات لحظة أو
 
Delegate Sub NoParameter() 
 
Sub Main() 
    Dim dlgt As NoParameter 
 
    dlgt = dlgt.Combine(New NoParameter(AddressOf MySub1), _ 
           New NoParameter(AddressOf MySub2)) 
 
    ' MySub2 و MySub1 سیتم تنفیذ كلا الاجرائين 
    dlgt.Invoke() 
End Sub 

  
 لذلك تمكنا من استدعائها قبـل  Shared Method طريقة مشتركة ()Combineالطريقة 

 إلى أن هـذه الطريقـة    أنبهكودعني هنا   .  السابق dlgtة جديدة من الكائن في المؤشر       إنشاء نسخ 
 ان كانـت  CType لذلك لابد من استخدام المعامـل  System.Delegateستعود بقيمة من النوع    

  : مفعلةOption Strict Onالعبارة 
 

dlgt = CType(dlgt.Combine( New NoParameter(AddressOf MySub1), _ 
       New NoParameter(AddressOf MySub2) ), NoParameter) 

  
 السابق، وعند اسـتدعاء الطريقـة       dlgtالمزيد أيضا، يمكن دمج إجراء ثالث في المؤشر         

Invoke()  فالأحدث الأقدم(سيتم تنفيذ جميع الإجراءات الثلاثة بالتتالي(:  
 
  ثالثإجراءدمج  '
dlgt = CType(dlgt.Combine(dlgt, New NoParameter(AddressOf MySub3)), _ 
       NoParameter) 
 
اتجراء تنفیذ الإسیتم '  
' MySub1() 
' MySub2() 
' MySub3() 
 بالتسلسل '
dlgt.Invoke() 

  
 Function حول دمج التفويضات، ففي حالة ما اذا كان نوع الإجراء المفوض        أخرىنقطة  

  :ود بالقيمة هو الذي سيعالأخير الإجراء أنفعليك معرفة 

Module Module1 
    Delegate Function NoParameterFun() As Boolean 
     
    Sub Main() 
        Dim dlgt As New NoParameterFun(AddressOf MyFunction1) 
        Dim dlgt2 As New NoParameterFun(AddressOf MyFunction2) 
 
        dlgt = CType(dlgt.Combine(dlgt, dlgt2), NoParameterFun) 
 
        ArabicConsole.WriteLine(dlgt.Invoke()) ' True 
    End Sub 
 
    Function MyFunction1() As Boolean 
        Return False 
    End Function 
 
    Function MyFunction2() As Boolean 
        Return True 
    End Function 
End Module 

  
 مـن   أجمـل  ولا   أسهل رغبت في حذف الإجراء من مؤشر التفويض، فلن تجد           إن،  أخيرا

  :()Removeالطريقة 
 
 

dlgt = CType(dlgt.Remove(dlgt, dlgt2), NoParameterFun) 
 
ArabicConsole.WriteLine(dlgt.Invoke()) ' False 
 
 

    ملاحظة
  

ــم إضــافت  ()Removeالطريقــة  ــر  الــسابقة تحــذف آخــر إجــراء ت ه للمتغی
 مع ذلك، جمیع الإجـراءات المدمجـة فـي المتغیـر تحفـظ فـي               .المفوض

لــذلك، الــشیفرة التالیــة .  وھــي مــصفوفةGetInvocationListالخاصــیة 
  :تمكنك من حذف الإجراء الاول ولیس الأخیر

 
dlgt.Remove(dlgt, dlgt.GetInvocationList(0)) 
ArabicConsole.WriteLine(dlgt.Invoke()) ' True 

  المواصفات
– جديـدة  NET Framework. عمل إطار التي يوفرها لك Attributesالمواصفات مصطلح 
 في لغات البرمجة، الهدف منها هو ان بعض محتويات مشاريعك قد لا تعتمد على كتابـة                 -نسبيا

 قد تشمل تعاريف ونـصوص حرفيـة   وإنما بشكل كامل، Source Codes الشيفرات المصدرية
Textual وعبارات توجيه للمترجم Compiler.  

 تعتمد على النصوص لتشرح الغرض منها       وأعضائهافمثلا، المعلومات العامة حول الفئات      
هناك أيضا بعض   . ومتطلباتها أو حتى الوصول إلى الصفحة المناسبة لملف التعليمات الخاص بها          

م في الشيفرات التي ترغب في  والتي تمكنك من التحكCompilerاوامر التوجيه الخاصة بالمترجم 
المزيد أيضا، قد تفرض وتحدد مواقع معينة لمتغيرات التركيبات . ترجمتها ان تحققت شروط معينة

  . وطريقة تحديد مواقعها بالذاكرةStructureمن نوع 
 متوفرة في لغات البرمجة ولكن لكل لغة طريقة خاصـة ومعينـة فـي               الأموربعض هذه   

 Visual للسيطرة على المترجم، واما في لغة pragma# العبارات ++C\Cتطبيقها، فنجد في لغة 

Basic 6 كـ ( فقد كان يعتمد بشكل كبير على صناديق الحوار التي تنشئ ملفات خاصة بهاctl .
، فقد توحدت فكـرة  NET Framework. عمل إطار مع  ولكن.لحفظ هذه المواصفات) idl.و 

 تمكنك من أصبحتليس هذا فقط، بل     . الأخرى NET.ت   بين لغا  Attributesتطبيق المواصفات   
  .بناء مواصفات خاصة بك

  
  Visual Basic .NETصیغة كتابة المواصفات في 

 ولكن لها صيغة قياسية لا بد من Classesهي الا فئات  في البداية عليك معرفة ان المواصفات ما  
 ـ لتتمكن من استخدامها في مشاريعك، استخدم علامة         إتباعها لكتابـة  >  و <  المعقوفـة  واسالأق

  :المواصفة
 
<System.ComponentModel.DefaultProperty ("XXX")> 

  
 باستيراد مجـال  اقترح عليكوبدلا من كتابة الاسم الكامل للمواصفة كما في المثال السابق،   

 لتختصر على نفسك كتابـة  ،Imports الذي توجد فيه المواصفة باستخدام     Namespace الأسماء
  : الكامل في كل مرة تود فيها استخدام المواصفةالاسم

 
Imports System.ComponentModel 
<DefaultProperty ("XXX")> 

 ستكتب المواصفة؟ والجواب يعتمد على نوع المواصفة التي ستـستخدمها،           أينالسؤال هنا   
 لآخـر افبعض المواصفات يمكن كتابتها في اي جزء أو كتلة من شيفراتك المصدرية، والـبعض        

 لـن   DebuggerStepThroughفمثلا، المواصفة   .  معينة كما سترى لاحقا    أماكنمحصور في   
  : عند تعريف الإجراءات فقطإلاتستطيع استخدامها 

 
<System.Diagnostics.DebuggerStepThrough()> Sub MySub() 
    … 
    … 
    … 
End Sub 

  
يفرة المصدرية بحيث تتوزع على سـطرين       لفصل المواصفة عن الش   _ قياسياً سنستخدم المعامل    

  :مختلفين لتسهيل قراءة وتتبع الشيفرة
 
<System.Diagnostics.DebuggerStepThrough()> _  
Sub MySub() 
    … 
    … 
    … 
End Sub 

  
  NET Framework. عمل إطارمواصفات من 

استخدامها فـي   العشرات من المواصفات والتي يمكن NET Framework. عمل إطاريوفر لك 
 متفرقة من البرنامج لأداء وظائف معينه، في الفقرات التالية سنلقي الضوء على بعض هذه               أجزاء

 لأعرض، كما سأخصص فقرة كاملة Visual Basic .NETالمواصفات والموجهة للغة البرمجة 
  .لك كيفية بناء مواصفة خاصة بك

  
  Conditional Attributeالمواصفة 

 ترجمة سطور معينة من الشيفرة المصدرية لحظة تحقق شرط معين، تقتضيرجم لمتاتوجيه عملية 
  : لهذا الغرضIf#في معظم لغات البرمجة السابقة يستخدم الموجه 

 
#If MyCondition Then 
    Sub MySub () 
        … 
        … 
    End Sub 
#End If 

Sub Main 
    MySub () 
    … 
    … 
    MySub () 
End Sub 

  
عـن طريـق    ( السابق   ()MySub الإجراء   إلغاء أردت إن تبرز   الأسلوبالمشكلة في هذا    

 فـي بـاقي سـطور       الأخطاءفستظهر مئات   ) False ليكون   MyConditionتغيير قيمة الشرط    
، احد الحلول هو استخدام الموجه  ()MySub عند كل استدعاء للإجراء      -لحظة الترجمة –البرمجة  

#Ifء عند كل استدعا:  
  

 
Sub Main 
    #If MyCondition Then 
        MySub () 
    #End If 
    … 
    … 
    #If MyCondition Then 
        MySub () 
    #End If 
    … 
    … 
End Sub 

  
صحيح ان الاسلوب السابق غير عملي، لذلك قد تدخل عبارة الشرط للموجه داخل الإجراء              

MySub()مة وتسهل عليك المه:  
 
Sub MySub () 
    #If MyCondition Then 
        … 
        … 
    #End If 
End Sub 

  
 فإن مئات السطور من الشيفرات  False تساوي   MyCondition كانت نتيجة الشرط     إنمع ذلك،   

 غير ذات قيمة، والتي ستتسبب      أصبحت)  بالتحديد ()MySubوالتي تستدعي الإجراء    (المصدرية  
  . إلى كبر حجم البرنامجبالإضافة -والتنفيذ أيضا–ية الترجمة في بطء عمل

 والتـي  Conditional Attribute المواصفة NET Framework. عمل إطاريوفر لك 
تمكنك من توجيه المترجم عند تحقق شرط معين، وطريقة استخدامها يكون عند تعريف الإجراءات  

  :Subمن نوع 
 
<Conditional("MyCondition")> _ 
Sub MySub() 
    … 
    … 
End Sub 
 
Sub Main 
    MySub () 
    … 
    … 
    MySub () 
    … 
    … 
End Sub 

  
، الأخـرى  في كافة ملفات المشروع      ه السابق وجميع السطور التي ستستدعي     ()MySubالإجراء  

 نتيجة الشرط سيتم تجاهلها لحظة الترجمة ولن يتم ترجمتها وتضمينها  في الملف النهائي ان كانت    
MyCondition تساوي صفر أو False.  

  
    ملاحظة

  
ــفة    ــق المواص ــك تطبی ــى الإConditionalلا يمكن ــراء عل ــوع  ج ــن ن ات م

Function.  
  

  
 السابق فما هو الا ثابت يمكنك تعريفه وتحديد قيمته مـن            MyConditionبالنسبة للشرط   

نتقال إلى خانـة التبويـب   ، والاProject Property Pagesصندوق حوار خصائص المشروع 
Configuration Properties ثم خانة التبويب الفرعية Build ثم كتابة اسم الثابت وقيمته تحت 

  .) بأعلى الصفحة المقابلة2-5شكل ( Custom Constantsخانة النص 
  

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


  الواجھات، التفويض، والمواصفات: الخامسالفصل   221  
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  

  
  .Project Property Pages تحديد ثوابت الترجمة في صندوق حوار :2-5شكل 

  
  DebuggerStepThrough Attributesالمواصفة 

 التي توفرهـا  Debugging سأتطرق إلى بعض طرق التنقيح الأخطاءاكتشاف  في الفصل السابع    
 مـن  Step Into الأمرلعمل ذلك اختر ( ومنها التنفيذ سطرا سطرا Visual Studio .NETبيئة 

  ]).F11[ أو اضغط على المفتاح Debugقائمة 
أكدا من صحتها ولست بحاجة إلى التدقيق فيها سطرا تلو  التي تكون متالإجراءاتفي بعض 

 عند بداية System.Diagnostics.DebuggerStepThrough، يمكنك استخدام المواصفة الآخر
  :الإجراء ليتم تنفيذه دفعة واحد

 
<System.Diagnostics.DebuggerStepThrough()> _ 
Sub MySub() 
    … 
    … 
    … 
End Sub 

  
  Obsolete Attributeالمواصفة 

 أو الخطأ لحظة ترجمة البرنامج،      Warning رسائل التنبيه    إظهارالغرض من هذه المواصفة هو      
 برنامجك إلى آخرين  وتود لفت انتباههم إلى جـزء  بإرساليمكنك الاستفادة منها مثلا عندما تقوم       

  :معين من الشيفرة أو التحقق منها

<Obsolete("  _ <(" شیفرة هذا الإجراء قبل التوزیعیا عباس، لا تنسى تغیير
Sub MySub() 
    … 
    … 
End Sub 

  
والتـي   (Output في نافذة المخرجات     Warning كإنذارعند الترجمة، ستظهر الرسالة السابقة      

  .)3-5شكل () View->Other Windows->Output الأمر باختيار إليهاتصل 
  

  
  . الترجمةثناءأ ظهرت Warning رسالة التنبيه :3-5شكل 

  
 أرسلت إن، Boolean من النوع أخرى تستقبل وسيطة   Obsoleteالمزيد أيضا، المواصفة    

 ستتحول الرسالة من رسالة تنبيه إلى رسالة خطأ لتوقف عملية الترجمة ولن يتم              Trueلها القيمة   
  :تنفيذ البرنامج

 
<Obsolete("الم اقل لك عدل الإجراء یا عنید", True)> _  
Sub MySub() 
    … 
    … 
    … 
End Sub 

  
    ملاحظة

  
 عند الـسطر الـذي    إلا خطأ، فلن يتم عرضھا      أوسواء كانت رسالة تنبیه     

  .Obsolete ولیس بمجرد استخدام المواصفة جراءيستدعي الإ

  FieldOffset والمواصفة StructLayoutالمواصفة 
، Structureى التركيبات من النـوع   موجه بشكل حصري إل    StructLayoutاستخدام المواصفة   

 القيمـة   أرسـل حيث تمكنك من تحديد طريقـة ترتيـب متغيـرات التركيـب فـي الـذاكرة،                 
LayoutKind.Auto للمعالجأداء وانسب أفضل للحصول على :  

 
 استيراد مجال الاسماء لاختصار الشیفرات '
Imports System.Runtime.InteropServices 
 
<StructLayout(LayoutKind.Auto)> _ 
Structure RGBValue 
    Dim Red As Byte 
    Dim Green As Byte 
    Dim Blue As Byte 
End Structure 

  
 فعليك تحديد موقع كل متغير من متغيرات التركيب LayoutKind.Explicit القيمة أرسلت إن أما

  :FieldOffsetفي الذاكرة بنفسك عن طريقة استخدام المواصفة 
 
<StructLayout(LayoutKind.Explicit)> _ 
Structure RGBValue 
    <FieldOffset(0)> Dim Red As Byte 
    <FieldOffset(1)> Dim Green As Byte 
    <FieldOffset(2)> Dim Blue As Byte 
End Structure 

  
 فـي    يعطيك مرونة كبيرة للتحكم في مواقع المتغيرات       FieldOffsetاستخدامك للمواصفة   

  :الذاكرة، فمثلا التركيب التالي لا يحتجز الا بايت واحد تصل اليه من ثلاث متغيرات
 
<StructLayout(LayoutKind.Explicit)> _ 
Structure RGBValue 
    <FieldOffset(0)> Dim Red As Byte 
    <FieldOffset(0)> Dim Green As Byte 
    <FieldOffset(0)> Dim Blue As Byte 
End Structure 

  
  :تأكد من كلاميتول

 
Dim X As RGBValue 
X.Blue = 10 
ArabicConsole.WriteLine(X.Red) ' 10 

المزيد أيضا، تستطيع الوصول إلى قيمة المتغيرات الثلاثة كلها عن طريق متغيـر واحـد،     
  : التركيب هذاراقب

 
<StructLayout(LayoutKind.Explicit)> _ 
Structure RGBValue 
    <FieldOffset(0)> Dim Red As Byte 
    <FieldOffset(1)> Dim Green As Byte 
    <FieldOffset(2)> Dim Blue As Byte 
    <FieldOffset(3)> Dim Extra As Byte 
    <FieldOffset(0)> Dim Value As Integer 
End Structure 

  
وهـو   (Value عن طريق المتغير     إماركيب   الحرية المطلقة في الوصول إلى قيمة الت       الآنلديك  

  :أو عن طريق كل متغير على حدة) الأربعةيشمل المتغيرات 
 
Dim X As RGBValue 
 
X.Value = 0 
ArabicConsole.WriteLine(X.Red) ' 0 
ArabicConsole.WriteLine(X.Green) ' 0 
ArabicConsole.WriteLine(X.Blue) ' 0 
 
X.Red = 255 
X.Green = 255 
X.Blue = 255 
ArabicConsole.WriteLine(X.Value) ' 16777215 

  
  
  . السابق في الذاكرةRGBValue يوضح لك مواقع متغيرات التركيب )4-5الشكل (
  

   
  . في الذاكرةRGBValue مواقع متغيرات التركيب :4-5شكل 

  بناء مواصفات خاصة
 تقليدية تحتوي على جميع Classes فئات إلا ما هي Custom Attributesالمواصفات الخاصة 

التي ترغب في ان تكون مواصفة -الشيفرات التي تود من المواصفة القيام بها، مع ذلك لابد للفئات 
Attributes-          ان تستوفي عدة شروط، سأذكرها تباعاً، وسنطبقها معاً على مراحل ليسهل عليـك 

  .استيعاب الفكرة
من الاحتفاظ بمجموعة بيانات تتعلق بالشيفرة      تمكنك   CodeInfoسأقوم معك ببناء مواصفة     

  ).الخ....كاسم كاتبها، اسم مراجعها، تاريخ كتابتها، (المصدرية 
  :Attributeيجب أن ينتهي اسم الفئة بالكلمة : الأولالشرط 

 
Class CodeInfoAttribute 
 
End Class 

  
  :System.Attributeان تكون الفئة مشتقة من الفئة : الشرط الثاني

 
Class CodeInfoAttribute 
    Inherits System.Attribute 
 
End Class 

  
 وتحديد المكان الذي يمكن استخدام      AttributeUsageضرورة استخدام المواصفة    : الشرط الثالث 

  :المواصفة الحالية في شيفرات البرنامج
 
<AttributeUsage(AttributeTargets.All) _ 
Class CodeInfoAttribute 
    Inherits System.Attribute 
 
End Class 

  
 المواصفة الحالية قابلة للاستخدام لأجعل وذلك AttributeTargets.Allتلاحظ اني ارسلت القيمة 

 لحـصر مجـال اسـتخدام هـذه        أخرى قيم   إرسال الشيفرات المصدرية، يمكنك     أماكنفي جميع   
 يوضح لك قيمة المواصفة، ومجال      يةالموجود في الصفحة التال   المواصفة في موقع معين، الجدول      

  .عملها

  المكان الذي ستستخدم فيه المواصفة الحالية  القيمة
AttributeTargets.All  في كل الاماكن.  

AttributeTargets.Assembly في المجمع الحالي.  
AttributeTargets.Class عند تعريف الفئات.  

AttributeTargets.Constructor مشيدات الفئات.  
AttributeTargets.Delegate  باستخدام الكلمة المحجوزة –الاجرائات المفوضة

Delegate.  
AttributeTargets.Enum  التركيبات من النوعEnum.  
AttributeTargets.Event حداثالأ.  
AttributeTargets.Field  الحقول.  

AttributeTargets.Interface الواجهات.  
AttributeTargets.Method الطرق.  
AttributeTargets.Module الوحدات البرمجية.  

AttributeTargets.Parameter اتجراءوسيطات الإ.  
AttributeTargets.Property الخصائص.  

AttributeTargets.ReturnValue  الاجرائات التي تعود بقيمةFunctions.  
AttributeTargets.Struct  التركيبات من النوعStructure.  

  
    ملاحظة

  
 لحظة ارسال القیمة Or من موقع باستخدام المعامل أكثريمكنك تحديد 

ــة تحــصر مجــال اســتخدام   AttributeUsageللمواصــفة  ، فالقیمــة التالی
  :المواصفة الحالیة على الطرق والخصائص فقط

  
<AttributeUsage(AttributeTargets.Method Or 
AttributeTargets.Property) 

  
  

، Boolean:  البيانات المسموح بها في هذا النوع مـن الفئـات هـي        أنواع: الرابعالشرط  
Byte  ،Short  ،Integer  ،Long  ،Char ،String ،Single ،Double ،Object التركيب من ،
  . فقطOne-Dimensional Array البعد أحادية، المصفوفات Enumنوع 

  :  المحددة في الشرط الرابعاعبالأنوسنقوم الآن بتعريف أعضاء الفئة مع الالتزام 
 

<AttributeUsage(AttributeTargets.All) _ 
Class CodeInfoAttribute 
    Inherits System.Attribute 
 
    Public ProgrammerName As String ' اسم المبرمج 
    Public TesterName As String ' اسم المراجع 
    Public Tested As Boolean  ' ت المراجعة؟تم  
 
    Sub New(ByVal programmerName As String) 
        Me.ProgrammerName = programmerName 
    End Sub 
End Class 

  
 وسيطة المشيد لحظة استخدامها فـي اي        أرسلالمواصفة جاهزة للاستخدام الان،     ! مبروك

  ):استدعاء المواصفة عند Attributeتجاهل كتابة الكلمة  لاحظ(مكان من البرنامج 
 
<CodeInfo("تركي العسيري")> _ 
Sub DoSomthing() 
    … 
    … 
End Sub 

  
، فيمكنك إسناد القـيم لهـا بهـذه         Tested و   TesterName الأخرىبالنسبة للخصائص   

  :الصيغة
 

<CodeInfo("تركي العسيري")> _ 
Class Person 
 
<CodeInfo("تركي العسيري", TesterName:=" السریععباس  ", Tested:=False)> _ 
    Sub Move () 
        … 
        … 
    End Sub 
 
<CodeInfo("تركي العسيري", TesterName:="عبود اللوح", Tested:=True)> _ 
    Sub Kill () 
        … 
        … 
    End Sub 
    … 
    … 
End Class 

 فـي   CodeInfoلحفظ بيانات المواصـفة     كنوع من الاقتراح، قد تضيف شيفرات إضافية        
  .ملفات نصية خارجية

  
  
  
  
  
  
  
  

 حقيقي بعـدما  Visual Basic .NET لتكون مبرمج أساسية قد انتهيت من تشييد بنية أكونبهذا 
 مـن  Visual Basic .NETتطرقت إلى جميع المبادئ والأساسيات التي لابد على كل مبـرمج  

 NET Framework. عمـل  إطـار استخدام مكتبة فئات  في برمجة وللإبحار وإتقانهامعرفتها 

Class Library أنـي  جزء ترغب به، رغم أي فإن لديك مطلق الحرية  في الانتقال إلى والآن 
الفئـات   بشدة بالالتزام  بالتسلسل الموضوع في هذا الكتاب والبدء من الفـصل التـالي                أنصحك

   قبل تطوير التطبيقات المتقدمةةالأولي حتى تتعرف على اكبر قدر من الفئات الأساسية

